clear

*Load Data
use "./data_geo.dta", clear

*****************************************************************************
*** Setup: see Kashyap & Stein (2000) and/or Den Haan, Sumner, Yamashiro (2002) for details ***

*Drop non-states 
drop if rssd9200 == "0"
keep if  rssd9210>0 & rssd9210< 57

*Drop Delaware and/or South Dakota (see Jayaratne & Strahan 1998 JLE)
*drop if rssd9210 == 10
*drop if rssd9210 == 46

*Sample period
keep if year_q < tq(1995q1)

*Positive Assets
keep if rcfd2170 > 0

*Commercial Banks Only
keep if rssd9048 == 200 

*Insured
*keep if rssd9424 == 1 | rssd9424 == 2 | rssd9424 == 6
drop if rssd9424 == 0

*Merger
drop if MERGE_CD != .

*Set panel data
xtset rssd9001 year_q

*Fix Total Loans pre-1984 (insert before calcuating loans2 above)
gen loans2 = rcfd1400
replace loans2 = rcfd1400+rcfd2165 if year_q < tq(1984q1)

*Put loans and loan components in real terms
gen c_p_i = cpi/100
gen r_loans = loans2/c_p_i 
gen r_ci = rcfd1600/c_p_i
gen r_rlest = rcfd1410/c_p_i
gen r_con = rcfd1975/c_p_i
gen r_ag = rcfd1590/c_p_i

*Log difference 
gen loans = ln(r_loans) - ln(L.r_loans)
gen ci = ln(r_ci) - ln(L.r_ci) 
gen re = ln(r_rlest) - ln(L.r_rlest) 
gen con = ln(r_con) - ln(L.r_con)
gen ag = ln(r_ag) - ln(L.r_ag) 

*Drop outliers
sort year_q
by year_q: egen sdv = sd(loans)
drop if loans >= 5*sdv
drop if loans <= -5*sdv

*Loan Component Outliers
gen ci_share = rcfd1600/loans2
*drop if ci_share < 0.05
gen re_share = rcfd1410/loans2
*drop if re_share < 0.05
gen con_share = rcfd1975/loans2
*drop if con_share < 0.05
gen ag_share = rcfd1590/loans2
*drop if ag_share < 0.05

*Four Consecutive Quarters of Loan Growth
*ssc install tsspell
tsspell, c(loans > 0)
sort rssd9001
by rssd9001: egen maxrun = max(_seq)
drop if maxrun < 4

*Percentiles
sort year_q
by year_q: egen p95 = pctile(rcfd2170), p(95)
by year_q: egen p99 = pctile(rcfd2170), p(99)

*Generate Quarter/Year/timetrend Series
sort rssd9001 year_q
gen Quarter = quarter(date2)
gen Year = year(date2) 
egen time = group(year_q)

*Choose to keep (or drop) year deregulation was implemented 
replace intra = 1 if intra == .
replace inter =1 if inter == .

* Generate Interaction Terms
gen rr_intra = rr*intra
gen rr_intra1 = L1.rr*intra
gen rr_intra2 = L2.rr*intra
gen rr_intra3 = L3.rr*intra
gen rr_intra4 = L4.rr*intra

gen rr_inter = rr*inter
gen rr_inter1 = L1.rr*inter
gen rr_inter2 = L2.rr*inter
gen rr_inter3 = L3.rr*inter
gen rr_inter4 = L4.rr*inter

gen ffr_intra = ffr_end*intra
gen ffr_intra1 = L1.ffr_end*intra
gen ffr_intra2 = L2.ffr_end*intra
gen ffr_intra3 = L3.ffr_end*intra
gen ffr_intra4 = L4.ffr_end*intra

gen ffr_inter = ffr_end*inter
gen ffr_inter1 = L1.ffr_end*inter
gen ffr_inter2 = L2.ffr_end*inter
gen ffr_inter3 = L3.ffr_end*inter
gen ffr_inter4 = L4.ffr_end*inter


*******************************************************************************
*** Table A.3 ***
gen inter_first2 = 0
replace inter_first2 = 1 if inter == 1 & intra == 0
sort rssd9001
by rssd9001: egen inter_first = max(inter_first2)

sort rssd9001 year_q
gen size = inter_first

gen size_intra = intra*size
gen size_inter = inter*size

gen rr_size = rr*size
gen rr_size1 = L1.rr*size
gen rr_size2 = L2.rr*size
gen rr_size3 = L3.rr*size
gen rr_size4 = L4.rr*size

gen rr_intra_size = rr*intra*size
gen rr_intra_size1 = L1.rr*intra*size
gen rr_intra_size2 = L2.rr*intra*size
gen rr_intra_size3 = L3.rr*intra*size
gen rr_intra_size4 = L4.rr*intra*size

gen rr_inter_size = rr*inter*size
gen rr_inter_size1 = L1.rr*inter*size
gen rr_inter_size2 = L2.rr*inter*size
gen rr_inter_size3 = L3.rr*inter*size
gen rr_inter_size4 = L4.rr*inter*size 

gen ln_size1 = L1.loans*size
gen ln_size2 = L2.loans*size
gen ln_size3 = L3.loans*size
gen ln_size4 = L4.loans*size

gen pi_size = pi_pchg*size
gen pi_size1 = L1.pi_pchg*size
gen pi_size2 = L2.pi_pchg*size
gen pi_size3 = L3.pi_pchg*size
gen pi_size4 = L4.pi_pchg*size

gen hpi_size = hpi_chg*size
gen hpi_size1 = L1.hpi_chg*size
gen hpi_size2 = L2.hpi_chg*size
gen hpi_size3 = L3.hpi_chg*size
gen hpi_size4 = L4.hpi_chg*size

egen size_yearq = group(size year_q)

reghdfe loans L(1/4).loans L(0/4).pi_pchg L(0/4).hpi_chg inter rr_inter rr_inter1 rr_inter2 rr_inter3 rr_inter4 size size_inter rr_size rr_size1 rr_size2 rr_size3 rr_size4 rr_inter_size rr_inter_size1 rr_inter_size2 rr_inter_size3 rr_inter_size4 ln_size1 ln_size2 ln_size3 ln_size4 pi_size pi_size1 pi_size2 pi_size3 pi_size4 hpi_size hpi_size1 hpi_size2 hpi_size3 hpi_size4, absorb(rssd9001 year_q size_yearq) cluster(rssd9210) 
lincom rr_inter + rr_inter1 + rr_inter2 + rr_inter3 + rr_inter4
lincom rr_size + rr_size1 + rr_size2 + rr_size3 + rr_size4
lincom rr_inter_size + rr_inter_size1 + rr_inter_size2 + rr_inter_size3 + rr_inter_size4
*******************************************************************************